

### Midterm Exam Solution

1. The formula =test(A1, A2) is placed in cell A3 and copied to cells B3:E3. The VBA code for function test is shown below. What are the resulting values in cells A3 to E3?

Function test(a As Double, b As Double) As Double

```

If a \ b > 0 Then
    test = a Mod b
ElseIf a / b < 1 Then
    test = a + b
Else
    test = 0
End If
End Function

```

	A	B	C	D	E
1	70	75	9	13	12
2	17	33	15	13	48
3					

	A	B	C	D	E
1	70	75	9	13	12
2	17	33	15	13	48
3	2	9	24	0	60

**Cell A3:** a = 70; b = 17; a\b = 70\17 = 4, which is > 0, so result in cell A3 is test = a Mod b = 70 Mod 17 = 2

**Cell B3:** a = 75; b = 33; a\b = 75\33 = 2 which is > 0 so result in cell B3 is test = a Mod b = 75 Mod 33 = 9

**Cell C3:** a = 9; b = 15; a\b = 9\15 = 0 which is not > 0 so we test 9/15 = 0.6 which is < 1 so result in cell C3 is test is a + b = 9 + 15 = 24.

**Cell D3:** a = 13; b = 13; a\b = 13\13 = 1 which is > 0, so result in cell D3 is test = 13 Mod 13 = 0.

**Cell E3:** a = 12; b = 48; a\b = 12\48 = 0 which is not > 0 so we test 12/48 = 0.25 which is < 1 so result in cell E3 is test is a + b = 12 + 48 = 60.

2. The formula that relates the recurring (periodic) loan payment, R, the amount of the loan (principal), P, the interest rate, i, and the number of payments) on the loan, N is shown below. Write a VBA function called LoanPayment that has P, i, and N as inputs and returns a value for R.

$$R = \frac{iP}{1 - (1+i)^{-N}}$$

Function LoanPayment(P As Double, i As Double, N As Integer) As Double

```

LoanPayment = P * i / (1 - (1 + i) ^ (-N))

```

End Function

3. The equation  $t = \frac{(-1)^n - 1}{2n + 1} \cos\left(\frac{(2n + 1)\pi x}{L}\right)$  is zero for even values of n. Assuming that all the variables have

been declared and the values of all variables on the right side of the equation have been defined, write only the if structure necessary to compute t from this equation, Your solution should set t = 0 for even n and make the correct computation for odd values of n.

```

If n Mod 2 = 0 Then

```

```

    t = 0

```

```

Else

```

```

    t = -2 * Cos((2 * n + 1) * PI * x / L) / (2 * n + 1)

```

```

End If

```

- 4 Trace all steps of the program shown below to determine the final value for P. (C, D, and P, are all type integer variables).

```

C = 13
D = 2
P = 1
Do while C > 0
  If C Mod 2 <> 0 Then
    P = P * D
    C = C - 1
  Else
    D = D^2
    C = C/2
  End If
Loop
MsgBox(P)

```

We enter the while loop because  $C = 13$  is  $> 0$ . The table below follows the progress through the loop. Trip is not part of this program; it is just a counter showing the number of times the while loop is used. "Initial C" is the value of C just before the while statement makes its  $C > 0$  test; "Exit C" is the value of C when the loop is exited. The attempt to make a 7<sup>th</sup> trip through the loop is halted because the  $C > 0$  test is false.

Trip	Initial C	C > 0	C Mod 2 <> 0	P	D	Exit C
1	13	True	True	2	2	12
2	12	True	False	2	4	6
3	6	True	False	2	16	3
4	3	True	True	32	16	2
5	2	True	False	32	256	1
6	1	True	True	8192	256	0
7	0	False				

Output P = 8192

5. List all the output from the following program segment. Show the reasoning you used in getting your answers.

```

Sub translate()
  Dim a As Integer, b As Integer, c As Integer, d As Integer
  a = 12
  b = 10
  c = 2
  For d = b To c Step -2
    If d Mod 3 = 0 Then
      a = a * d
      MsgBox ("At d = " & d & ", a = " & a)
    Else
      a = a - b
    End If
  Next d
End Sub

```

The for loop will execute for  $d = 10, 8, 6, 4,$  and  $2$ . After the loop is completed for  $d = 2$ ,  $d$  is decremented to  $0$  and the condition  $d > c$  is no longer met so the loop exits. We get output only when the if condition is true; that is only when  $d$  is evenly divisible by  $3$ . This occurs only one time, when  $d = 6$ . We have the following calculations in the for loop.

$d = 10$ ( $d \text{ Mod } 3 = 0$ is false)	$a = a - b$ or $a = 12 - 10 = 2$
$d = 8$ ( $d \text{ Mod } 3 = 0$ is false)	$a = a - b$ or $a = 2 - 10 = -8$
$d = 6$ ( $d \text{ Mod } 3 = 0$ is true)	$a = a * d$ or $a = (-8)(6) = -48$ . The only output occurs here: <b>At d = 6 a = -48</b>
$d = 4$ ( $d \text{ Mod } 3 = 0$ is false)	$a = a - b$ or $a = -48 - 10 = -58$
$d = 2$ ( $d \text{ Mod } 3 = 0$ is false)	$a = a - b$ or $a = -58 - 10 = -68$
$d = 0$ ( $d > c = 2$ is false so loop terminates)	

6. Electrical resistors have three colored bands where each color represents an integer code according to the table below

color	black	brown	red	orange	yellow	green	blue	violet	gray	white
code value	0	1	2	3	4	5	6	7	8	9

The resistance (in ohms) is found as follows: (1) multiply the code value corresponding to the first band by 10. (2) Add the result to the code value for the second band. (3) Multiply the sum found in step 2 by  $10^n$ , where n is the code value of the third band. For example if the bands are red, yellow and orange (with code values 2, 4, and 3) the resistance would be  $(2*10+4)*10^3 = 24000$  ohms.

- a. Write a VBA function `getValue` which has the name of the color as a string input and returns the numerical value.

	A	B	C
1	Resistance Color Calculator		
2	Tens	Units	$10^n$ Factor
3	red	yellow	violet
4			
5	R =	2.4E+08	ohms
6	Find Resistance		
7			
8			

```
Function getValue(color As String)
    If color = "black" Then
        getValue = 0
    ElseIf color = "brown" Then
        getValue = 1
    ElseIf color = "red" Then
        getValue = 2
    ElseIf color = "orange" Then
        getValue = 3
    ElseIf color = "yellow" Then
        getValue = 4
    ElseIf color = "green" Then
        getValue = 5
    ElseIf color = "blue" Then
        getValue = 6
    ElseIf color = "violet" Then
        getValue = 7
    ElseIf color = "gray" Then
        getValue = 8
    ElseIf color = "white" Then
        getValue = 9
    End If
End Function
```

- b. The Excel Resistance Color calculator shown at the right can be used to find the resistance from a pull-down menu of colors in cells A3:C3. Write the macro that takes color data from cells A3:C3, computes the resistance using the function you wrote in part a, and writes the resistance value to cell B5.

```
Sub calculateResistance()
    Dim unitsColor As String
    Dim tensColor As String
    Dim tenToTheNthColor As String
    Dim units As Integer
    Dim tens As Integer
    Dim tenToTheNth As Integer
    unitsColor = Range("B3").Value
    tensColor = Range("A3").Value
    tenToTheNthColor = Range("C3").Value
    units = getValue(unitsColor)
    tens = getValue(tensColor)
    tenToTheNth = getValue(tenToTheNthColor)
    Range("B5").Value = (tens * 10 + units) * 10 ^ tenToTheNth
End Sub
Sub calculateResistanceAlternateNotRecommendedCode()
    Range("B5").Value = ((getValue(Range("A3").Value)) * 10 +
        (getValue(Range("B3").Value))) * 10 ^ getValue(Range("C3").Value))
End Sub
```